
Curry-Howard対応
1 プログラムの型付け
型 τ, θ ::= nat | Z | . . . | θ → τ | τ × θ データ型, 関数型, 直積

型判定 Γ ⊢ M : τ Γ = x1 : τ1, . . . , xn : τn という仮定のもとで，M が型 τ をもつ．

型付け規則 Coqの式は以下の型付け規則によって型付けされる．

変数 Γ ⊢ x : τ (x : τはΓに含まれる)

抽象 Γ, x : θ ⊢ M : τ

Γ ⊢ fun x:θ ⇒ M : θ → τ

適用 Γ ⊢ M : θ → τ Γ ⊢ N : θ

Γ ⊢ M N : τ

定義 Γ ⊢ M : θ Γ, x : θ ⊢ N : τ

Γ ⊢ let x := M in N : τ

不動点 Γ, f : θ → τ, x : θ ⊢ M : τ

Γ ⊢ fix f (x:θ) := M : θ → τ

直積　 Γ ⊢ M : τ Γ ⊢ N : θ

Γ ⊢ (M,N) : τ × θ

射影 Γ ⊢ fst : τ × θ → τ Γ ⊢ snd : τ × θ → θ

実際のプログラムの型付けには単一化に基いた型推論が使われている．

型付けの例
Γ, x : nat ⊢ S : nat → nat Γ, x : nat ⊢ x : nat

Γ, x : nat ⊢ S x : nat

Γ ⊢ fun x:nat ⇒ S x : nat → nat
抽象

適用
Γ ⊢ O : nat

Γ ⊢ (fun x:nat ⇒ S x) O : nat
適用

2 命題と型の対応

カリー・ハワード同型により，1回目に見た命題論理と型理論 (型付λ計算)が対応している．

公理　 ∆ ⊢ P (P は∆に含まれる)

→導入 ∆, P ⊢ Q

∆ ⊢ P → Q

→除去 ∆ ⊢ P ∆ ⊢ P → Q

∆ ⊢ Q

矛盾　 ∆ ⊢ False

∆ ⊢ P

∧導入 ∆ ⊢ P ∆ ⊢ Q

∆ ⊢ P ∧Q

∧除去 ∆ ⊢ P ∧Q

∆ ⊢ P

∆ ⊢ P ∧Q

∆ ⊢ Q

∨導入 ∆ ⊢ P

∆ ⊢ P ∨Q

∆ ⊢ Q

∆ ⊢ P ∨Q

∨除去 ∆ ⊢ P ∨Q ∆, P ⊢ R ∆, Q ⊢ R

∆ ⊢ R

具体的には，以下のような対応が見られる．

命題 (論理式) 型
証明 (導出) プログラム (証明項)

仮定∆ 型環境 Γ

→ →
∧ ×
∨ +

導出規則と型付け規則も基本的には 1対 1で対応している．それぞれの体系を少し修正すると
以下の定理がなりたつ．

定理 1 (Curry-Howard同型) ある同型 ⟨ ⟩ :命題→型が存在し，任意の∆と P について，導
出 Πより∆ ⊢ P が示せるならば，ΠからプログラムM が作れ，⟨∆⟩ ⊢ M : ⟨P ⟩．また，任意の
Γ,M, τ について型理論で Γ ⊢ M : τ が導出できれば，命題論理において ⟨Γ⟩−1 ⊢ ⟨τ⟩−1が導出で
きる．

ただし，上の不動点の規則は気を付けないと矛盾を生んでしまう．具体的には，θ = True と
τ = False にすると，以下の導出が可能になる．

Γ , f : True → False, x : True ⊢ f x : False

Γ ⊢ fix f (x :θ) := f x : True → False

それを考慮したCoqの本当の不動点の規則はさらに f が xより小さな引数に適用されることを求
めているので，この矛盾が実際には起きない．本当の規則が複雑なのでここには書かない．
この関係は述語論理や高階述語論理に拡張できる．ただし，論理演算子がそのまま型として使

われるので，重要なのは証明項の存在である．特に，∃x : T, P (x) の証明項は型 T の元 tと P (t)

の証明項の対になる．
Coqの場合では，Inductiveの表現力が高く，「→」以外の各論理演算子が Inductiveとして定義

され，除去規則は match構文を使った証明項に翻訳される．∧，∨，Falseや ∃の除去がその例．

Require Import ssreflect.

Section CurryHoward.

Variables P Q R : Prop.

Goal P -> (P -> Q) -> Q.

move=> p pq. apply pq. assumption.

Show Proof.

(fun (p : P) (pq : P -> Q) => pq p)

Abort.

Goal P /\ Q -> Q /\ P.

move=> pq; case pq. split; assumption.

Show Proof.

(fun pq : P /\ Q => match pq with

| conj H H0 => conj H0 H

end)

Abort.

Goal P \/ Q -> (P -> R) -> (Q -> R) -> R.

move=> pq pr qr. case pq; assumption.

Show Proof.

(fun (pq : P \/ Q) (pr : P -> R) (qr : Q -> R) =>

match pq with

| or_introl x => pr x

| or_intror x => qr x

end)

Abort.

Goal False -> P.

move=> f; case f.

Show Proof.

(fun f : False => match f return P with end)

Abort.

Variables (A : Set) (drinks : A -> Prop).

Variable me : A. (* 自分がそこにいる *)

Hypothesis EM : forall P, P \/ ~ P. (* 排中律 *)

(* 飲み屋には必ずそいつが飲めば全員は飲むというやつがいる *)

Theorem drinkers_paradox : exists x, drinks x -> forall y, drinks y.

Proof.

case (EM (exists x, ~drinks x)) => H.

case H => x Hx. (* 飲まないやつを見付けたので，そいつにする *)

exists x => Hnx; case Hx; assumption.

exists me => _ y. (* 皆飲んでいるので，誰でもいい *)

case (EM (drinks y)) => Hy.

assumption.

case H; exists y; assumption.

Show Proof.

match EM (exists x : A, ~ drinks x) with

| or_introl (ex_intro _ x Hx) =>

ex_intro (fun x0 : A => drinks x0 -> forall y : A, drinks y) x

(fun Hnx : drinks x => (* 矛盾 *)

match Hx Hnx return (forall y : A, drinks y) with end)

| or_intror H =>

ex_intro (fun x : A => drinks x -> forall y : A, drinks y) me

(fun (_ : drinks me) (y : A) =>

match EM (drinks y) with

| or_introl Hy => Hy

| or_intror Hy => (* 矛盾 *)

match H (ex_intro (fun x : A => ~ drinks x) y Hy) return (drinks y)

with end

end)

end

Qed.

End CurryHoward.

3 eqTypeと seq

Mathcomp ではある型 Tにおける等価性が判定可能ということをソートみたいに表現できる．
2つの Tの値が等価かどうかを計算する T -> T -> boolの関数があることを T : eqTypeと宣
言できる．そうすれば，aと bの型が Tなら，a == bがその関数を呼ぶ．
実は eqTypeは Typeや Setのようなソートではなく，その関数とその正しさの証明を含むデー

タ構造である．
Record eqType : Type := Pack

{ T : Type; op : T -> T -> bool; _ : forall x y, reflect (x = y) (op x y) }

ただし，たとえば nat : eqTypeと書いても，本当は nat : Set なのでそのまま eqTypeの値で
はないので，あらかじめ宣言された nat eqTypeが自動的に使われ，区別する必要があまりない．
MathCompのリストの関数はモジュール seqで定義されている．因に，その中で Notation seq

:= listと宣言されているので，MathComp を使うと全てのリストの型が seqと表示される．
リストに関する関数は等価性の判定が必要なもの以外は普通に一般的な Typeに対して定義さ

れているが，各関数の性質を表した述語などがほとんどの場合 eqTypeを仮定している．特に，あ
る元 t : T : eqTypeがリスト l : seq Tに含まれることを t \in lと書く．
また，ビュー機構を使い，bool 上の述語と量化子を使った表現を結びつけている．たとえば，

全称は，
Variables (T : eqType) (a : T -> bool).

Fixpoint all s := if s is x :: s’ then a x && all s’ else true.

Lemma allP s : reflect (forall x : T, x \in s -> a x) (all a s).

4 整列の証明, 再び
Section sort.

Variable A : eqType.

Variable le : A -> A -> bool.

Variable le_trans: forall x y z, le x y -> le y z -> le x z.

Variable le_total: forall x y, ~~ le x y -> le y x.

Fixpoint insert a l := match l with

| nil => (a :: nil)

| b :: l’ => if le a b then a :: l else b :: insert a l’

end.

Fixpoint isort l :=

if l is a :: l’ then insert a (isort l’) else nil.

Fixpoint sorted l := (* allを使って bool上の述語を定義する *)

if l is a :: l’ then all (le a) l’ && sorted l’ else true.

Lemma le_seq_insert a b l :

le a b -> all (le a) l -> all (le a) (insert b l).

Proof.

elim: l => /= [-> // | c l IH].

move=> leab /andP [leac leal].

case: ifPn => lebc /=.

- by rewrite leab leac.

- by rewrite leac IH.

Qed.

Lemma le_seq_trans a b l :

le a b -> all (le b) l -> all (le a) l.

Proof.

move=> leab /allP lebl.

apply/allP => x Hx.

by apply/le_trans/lebl.

Qed.

Theorem insert_ok a l : sorted l -> sorted (insert a l). Admitted.

Theorem isort_ok l : sorted (isort l). Admitted.

(* perm_eq が seq で定義されているが補題だけを使う *)

Theorem insert_perm l a : perm_eq (a :: l) (insert a l).

Proof.

elim: l => //= b l pal.

case: ifPn => //= leab.

by rewrite (perm_catCA [:: a] [:: b]) perm_cons.

Qed.

perm_eq_trans : forall (T : eqType), transitive (seq T) perm_eq

Theorem isort_perm l : perm_eq l (isort l). Admitted.

End sort.

Check isort.

Definition isortn : seq nat -> seq nat := isort _ leq.

Definition sortedn := sorted _ leq.

Lemma leq_total a b : ~~ (a <= b) -> b <= a. Admitted.

Theorem isortn_ok l : sortedn (isortn l) && perm_eq l (isortn l). Admitted.

Require Import Extraction.

Extraction "isort.ml" isortn. (* コードが分かりにくい *)

OCamlで正しく動く．
% ocamlc -c isort.mli isort.ml

% ocaml

OCaml version 4.07.1

# #load"isort.cmo";;

# open Isort;;

# isortn (Cons (S O, Cons (O, Cons (S (S O), Nil))));;

- : Isort.nat Isort.list = Cons (O, Cons (S O, Cons (S (S O), Nil)))

練習問題 4.1 Admitted を Proof に変え，証明を完成させよ．

5 偶奇，mathcomp式
Propでの述語を使ったとき，述語の導出に対する帰納法の重要性を見た．しかし，mathcomp

式に述語を boolで定義すると，通常の帰納法で足りることが多い．
odd は ssrnat で boolの述語とした定義されている．even をその否定として定義すると様々

なことが簡単になる．

Section even_odd.

Notation even n := (~~ odd n). (* 単なる表記なので，展開が要らない *)

Theorem even_double n : even (n + n).

Proof. elim: n => // n. by rewrite addnS /= negbK. Qed.

(* 等式を使って nに対する通常の帰納法を可能にする *)

Theorem even_plus m n : even m -> even n = even (m + n).

Proof.

elim: n => /= [|n IH] Hm.

- by rewrite addn0.

- by rewrite addnS IH.

Qed.

Theorem one_not_even : ~~ even 1.

Proof. reflexivity. Qed.

Theorem even_not_odd n : even n -> ~~ odd n.

Proof. done. Qed.

Theorem even_odd n : even n -> odd n.+1. Admitted.

Theorem odd_even n : odd n -> even n.+1. Admitted.

Theorem even_or_odd n : even n || odd n. Admitted.

Theorem odd_odd_even m n : odd m -> odd n = even (m+n). Admitted.

End even_odd.

練習問題 5.1 Admitted を Proof に変え，証明を完成させよ．

